

Zeitgleichungs-Berechnung für 1800 - 2200

Helmut Haase*

28. August 2013

Erste Fassung: 18. Mai 2012

Teil von: [Scilab-Lernseite des Autors](#)

Zusammenfassung

Ausgehend vom Keplerschen Bahnmodell des Systems Erde-Sonne wird ein Formelsatz zur Berechnung der Zeitgleichung angegeben. Letztere bildet den ungleichförmigen Gang der Sonne am Himmel ab (Kurvenverlauf z. B. für das Jahr 2000 s. Abb. 1).

Die Formeln liefern genaue Werte für mehrere Jahrhunderte. Im Zeitraum 1800 - 2050 betragen die maximale und die mittlere Abweichung gegenüber einem Referenzmodell, das auch die Mond- und Planetenkräfte abbildet, 2,5 bzw. 0,5 Sekunden. Alle erforderlichen Parameterwerte sind angegeben. Ergänzende Daten aus Jahrbüchern werden nicht benötigt.

Für Leser mit Vorkenntnissen auf dem Gebiet der Himmelmechanik genügt es, nur den Anhang C zu lesen.

Inhaltsverzeichnis

1. Einführung	3
2. Definition	3
3. Modell und Koordinaten	3
4. Zeitskalen und säkulare Änderungen der Parameter	4
5. Mittlere Rektaszension	6
6. Wahre Rektaszension	7
7. Zeitgleichung	7
8. Synchronisation mit dem Kalender	7
9. Exzentrische und wahre Anomalie	8
10. Ekliptische Länge	8
11. Genauigkeit	8
12. Literatur und Web-Links	8
A. Übergang vom helio- zum geozentrischen Bezugssystem	9
B. Programm-Code	9
C. Berechnungsvariante mit Zeitbezug aus mittlerer Länge	10

*helmut.haase@TELOS-Web.de Fehlerhinweise, Verbesserungsvorschläge oder Kommentare sind willkommen.




Abbildung 1: Zeitgleichung des Jahres 2000

Die t -Werte am unteren Bildrand zählen ab Frühlingsanfang. Mit dieser Skalierung setzt sich die Zeitgleichung in den Jahren davor und danach in guter Näherung periodisch fort – nicht exakt, weil sich ihre Parameter mit der Zeit geringfügig ändern. Die untere Skalierung betont das Charakteristische der Kurve, während sich die oben eingetragenen Monatsanfänge Jahr für Jahr gegen die Kurvenform verschieben. Die Marken F, S, H und W in Bildmitte stehen für die Jahreszeit-Anfänge auf der Nordhalbkugel, P und A für Perihel bzw. Aphel.

1. Einführung

Wenn die Fixsterne in der scheinbaren Umgebung der Sonne am Himmel für das Auge sichtbar wären, ließe sich direkt beobachten, wie die Sonne beim täglichen Lauf von Osten nach Westen etwa 1° oder zwei Sonnendurchmesser gegenüber den Fixsternen zurückbleibt, was sich im Laufe eines Jahres auf 360° aufsummiert. Der diese Relativbewegung abbildende Winkel ist die Rektaszension¹. Die Rektaszension der Sonne verläuft ungleichmäßig: Wenn man ihre zeitlich lineare – d. h. gleichmäßige – Komponente abzieht, verbleibt eine überlagerte Schwankung in der Größenordnung von ± 4 Grad oder ± 16 Minuten, die durch die Zeitgleichung g beschrieben wird.

Wäre die Erdbahn kreisförmig und die Erdachse normal zur Ekliptik, wäre die Zeitgleichung überflüssig. Sie hätte dauernd den Wert null. Anders ausgedrückt: Die Exzentrizität der Erdbahn und die Schiefe der Äquator- gegen die Ekliptik-Ebene verursachen die Zeitgleichung, wobei der Beitrag der Schiefe den der Exzentrizität überwiegt.

2. Definition

Die Zeitgleichung $g(t)$ bildet den ungleichförmigen Gang der Sonne am Himmel ab. Sie ist nach [Schneider] durch die Rektaszensionsdifferenz

$$g(t) = \alpha_m(t) - \alpha(t) \quad (1)$$

definiert. Darin bezeichnet t die gleichmäßig fortschreitende (Skalen-)Zeit², α die am Himmel messbare Rektaszension der (wahren) Sonne und α_m die Rektaszension einer fiktiven mittleren Sonne. Letztere läuft als theoretische Vergleichssonne in derselben Umlaufzeit T_{tr} wie die wahre Sonne, aber mit konstanter Winkelgeschwindigkeit auf dem Himmelsäquator so um, dass sich die wahre und die mittlere Sonne während eines Jahres viermal gegenseitig überholen. Der auf ein Winkelgrad gerundete Betrag von g liegt dabei, außer gegen Novemberanfang, unter 4° .

Die Zeitgleichung kann, wie die Rektaszension, als Zeit (z. B. in Stunden) oder als Winkel (z. B. in Grad oder Bogenmaß) angegeben werden. Dabei entspricht der Vollwinkel 24 h.

Mit der Definition laut (1) lässt sich aus der wahren (aus dem Sonnenstand abgeleiteten) Ortszeit WOZ ³ die gleichmäßig ablaufende mittlere Ortszeit

$$MOZ = WOZ - g \quad (2)$$

bestimmen.⁴

Genauere Werte der Zeitgleichung sind astronomischen Jahrbüchern zu entnehmen oder mit Computerprogrammen zu berechnen, welche die Einflüsse von Mond und Planeten berücksichtigen. Hier wird eine Näherung von g auf Basis des (Zwei-Massen-)Kepler-Modells erläutert.

3. Modell und Koordinaten

Im Folgenden ist ein Rechenweg dargestellt, wie der Zeitgleichungs-Wert für einen bestimmten Zeitpunkt D (z. B. 1.2.2000 12:30:45 UTC) bestimmt werden kann.

Das zugrunde gelegte Keplersche Zwei-Massen-Modell berücksichtigt nur die Gravitationskraft zwischen Sonne und Erde. Die Einflüsse von Mond und Planeten sind nicht enthalten. Der Ansatz reicht trotzdem aus, um g -Werte mit einer Abweichung von weniger als 3 s gegenüber genauen Messungen zu berechnen.

Die folgende Herleitung der Zeitgleichung geht von der Beschreibung der Sonnenbahn in zwei geozentrischen astronomischen [Koordinatensysteme]n aus (Abb. 2). Das abgebildete Modell (Sonne auf ihrer scheinbaren Ellipsenbahn) ist kinetisch (die Kräfte betreffend) unhaltbar,⁵ kinematisch

¹ Die Rektaszension eines Gestirns ist der Winkel zwischen dem (auf dem Himmelsäquator liegenden) Frühlingspunkt und dem Schnittpunkt des Himmelsäquators mit dem Stundenkreis des Gestirns – hier der Sonne. Der Frühlingspunkt kann als fiktiver Fixstern aufgefasst werden. Die Rektaszension eines Fixsterns ist konstant.

² Über den Nullpunkt der Skalenzeit kann verfügt werden. Weiter unten wird dafür der Frühlingsanfang des Jahres 2000 gewählt.

³ Wenn die Sonne an einem Ort im Süden steht, ist dort und auf dem Meridian des Ortes 12:00 WOZ.

⁴ Die MOZ schreitet wie t gleichmäßig fort. Nach MOZ oder t gemessene Zeitspannen sind gleich. Das gilt für WOZ -Zeitspannen nicht. Z. B. liegt zwischen einem Ereignis Mitte Februar und einem zweiten *zur gleichen* MOZ Anfang November danach (oder späterer Jahre) nicht eine ganzzahlige Anzahl von 24-h-Tagen, sondern etwa eine halbe Stunde weniger.

⁵ ... es sei denn, Sonne und Erde tauschten ihre Massen.




Abbildung 2: Sonne auf ihrer scheinbaren Bahn im geozentrisch-ekliptischen (grün, x, y, z) und geozentrisch-äquatorialen Koordinatensystem (blau, x', y', z').
 F, S, H, W: Sonnenpositionen bei den vier Jahreszeit-Anfängen auf der Nordhalbkugel. P, A: Perihel bzw. Aphel. Die beiden farbigen Dreiecke sind Ausschnitte aus den jeweiligen Ebenen und dienen zur Herleitung der Umrechnungsformel zwischen ekliptischer Länge Λ und (äquatorialer) Rektaszension α auf Basis der Strecken ℓ , ℓ_x , ℓ_y und $\ell_{y'}$. Die zum Frühlingpunkt gerichtete x -Achse ist beiden Systemen gemeinsam. Um diese Achse ist das (x', y', z') -System um den Schiefewinkel ε gegen das (x, y, z) -System gedreht. Die z' -Achse – identisch mit der Erdachse – zeigt zum Himmelsnordpol, Υ bezeichnet die Frühlingpunkt-Richtung. Weitere Symbole, speziell ϖ , s. Tab. 1! Im Raum parallele Linien sind auch im Bild parallel. Das entsprechende heliozentrische Bild unterscheidet sich hauptsächlich dadurch, dass Sonne und Erde ihre Plätze tauschen und der Υ -Pfeil entgegengesetzt gerichtet ist. Siehe hierzu auch Anhang A!

(die zeitabhängigen Positionen, Entfernungen und Winkel betreffend) dagegen korrekt. Insbesondere kann der aus dem Kepler-Modell bekannte Formelsatz für Winkelgrößen übernommen werden, der üblicherweise in heliozentrisch-ekliptischen Koordinaten dargestellt wird: Der geozentrische Ortsvektor der Sonne ist dem heliozentrischen Ortsvektor der Erde genau entgegengerichtet. Die Vektoren unterscheiden sich nur im Vorzeichen! Der hier verwendete Formelsatz zum Kepler-Modell ist im Wikipedia-Artikel [\[Kepler-Gleichung\]](#) nachzulesen.

4. Zeitskalen und säkulare Änderungen der Parameter

In den Formeln weiter unten erscheinen drei Zeitskalen mit gleichmäßig fortlaufender Zeit, die sich durch ihren Nullpunkt unterscheiden:

- Die Skalenzeiten t und t' haben ihren Nullpunkt beim Frühlingsanfang des Jahres 2000 bzw. beim Periheldurchgang. Beide werden in Tagen (d) angegeben.
- Die Säkularezeit T rechnet ab der Standarddepoche J2000.0, d.h. ab 1.1.2000 12:00 UTC und wird in Jahrhunderten angegeben (1 Jahrhundert = 36525 d).

⁶Heliozentrische und geozentrische Längen unterscheiden sich um 180° , vgl. Schluss von Abschn. 3.

Nr.	Symbol	Bezeichnung	Wert
1	D	Datum, für das g berechnet wird	1. Februar 2000, 12:30:45 UTC
2	D_F	Frühlingsanfangs-Datum	20. März 2000, 7:35 UTC
3	e_2	numerische Exzentrizität der Umlaufbahn	0.016709
4	\dot{e}_2	zeitliche Änderung von e	$4.392 \cdot 10^{-5}$ /Jahrhundert
5	ε_2	Schiefe der Ekliptik	$23^\circ 26' 21,4''$
6	$\dot{\varepsilon}_2$	zeitliche Änderung von ε	$-46.8''$ /Jahrhundert
7	ϖ_2	heliozentrische Länge ⁶ des Perihels	102.93735°
8	$\dot{\varpi}_2$	zeitliche Änderung von ϖ , Apsidendrehung	1.72° /Jahrhundert
9	T_{tr}	Umlaufperiode, tropisches Jahr	365.242190 d
10	t	Zeit bei D ab D_F	-47.794618 d
11	t_2	Zeit bei Epoche J2000.0	-78.815972 d
12	T	Säkularzeit	0.00084932 Jahrhundert
13	e	numerische Exzentrizität bei D	0.01670897
14	ε	Schiefe bei D	23.43927°
15	ϖ	helioz. Länge des Perihels bei D	102.93881°
16	n_{tr}	mittlere Winkelgeschwindigkeit des Sonnenumlaufs	0.98565° /d
17	V_F	wahre Anomalie bei D_F	77.06119°
18	E_F	exzentrische Anomalie bei D_F	76.12984°
19	t'_F	P-Zeit (Zeit ab Perihelpassage) bei D_F	76.295444 d
20	t'	P-Zeit bei D	28.500826 d
21	M	mittlere Anomalie bei D	28.09176°
22	α_m	Rektaszension der mittleren Sonne bei D	-48.96943°
23	E	exzentrische Anomalie bei D	28.54930°
24	V	wahre Anomalie bei D	29.01024°
25	Λ	geoz. Länge bei D	-48.05095°
26	α	Rektaszension der wahren Sonne bei D	-45.58947°
27	g	Zeitgleichung bei D	-0.0589913 rad
28		Zeitgleichung bei D	-3.37995°
29		Zeitgleichung bei D	-13.52 Minuten
30	Υ	Zeichen für Frühlingspunkt-Richtung	

Tabelle 1: Zusammenstellung der verwendeten Größen und Formelzeichen.

Nrn. 1 und 10: Die unabhängige Variable in Datums- bzw. Skalenzeit-Form. Nr. 2: D_F kann viele Jahre entfernt von D liegen. Die Nrn. 2 bis 9 enthalten die konstanten Parameter der Zeitgleichung (gültig für 1800 bis 2200), die Nrn. 10 bis 29 die berechneten Zahlenwerte für das Beispiel-Datum D (s. Anhang B). Der Index 2 bei den Parametern weist auf die Standardepoche J2000.0 hin.

Bei der Umrechnung eines bestimmten Datums D (z. B. 1.2.2000 12:30:45 UTC) in t -Zeit kann ein Tabellenkalkulationsprogramm oder Mathematiksoftware helfen. Darin steht normalerweise eine Funktion zur Verfügung, hier mit d bezeichnet, die ein Datum in eine fortlaufende Zeit in Tagen umrechnet. Dabei spielt der Nullpunkt der programminternen Skala keine Rolle, da nur Zeitdifferenzen auszurechnen sind. Für die Zeit ab Frühlingsanfang⁷ erhält man

$$t = d(D) - d(D_F) \quad (3)$$

und für die Säkularzeit (Jahrhundertzeit)

$$T = \frac{d(D) - d(1.1.2000 \ 12:00 \ \text{UTC})}{36525}. \quad (4)$$

Die Zeit ab Periheldurchgang t' ist die Zeit des Kepler-Modells, T dient zur Korrektur der langsam veränderlichen Parameter der Rechnung. Das Nullpunktsdatum im Zähler von (4) steht für die Standarddepoche J2000.0.

Die langsamen Änderungen der Parameter e , ε und ϖ werden durch die zeit-linearen Ansätze

$$e = e_2 + \dot{e}_2 T \quad (5)$$

$$\varepsilon = \varepsilon_2 + \dot{\varepsilon}_2 T \quad (6)$$

bzw.

$$\varpi = \varpi_2 + \dot{\varpi}_2 T \quad (7)$$

berücksichtigt (Bezeichnungen s. Tab. 1). Dadurch gilt der hier angegebene Formelsatz zur Zeitgleichung für mehrere Jahrhunderte. Die Zahlenwerte sind $[e_2, \dot{e}_2, \varpi_2]$, $[\dot{\varpi}_2]$, $[\varepsilon_2, \dot{\varepsilon}_2]$ und $[T_{tr}]$ entnommen.

5. Mittlere Rektaszension

Die gleichmäßig mit der Zeit zunehmende mittlere Rektaszension α_m , die als Vergleichsgröße in (1) dient, ist der geozentrische Winkel der fiktiven mittleren Sonne gegen die Frühlingspunkt-Richtung Υ im geozentrisch-äquatorialen Koordinatensystem. Letzteres entsteht aus dem geozentrisch-ekliptischen System (Abb. 2) durch linkshändige Drehung um die x-Achse um den Winkel ε . Zur P-Zeit (s. Tab. 1) des Periheldurchgangs der wahren Sonne, d. h. bei $t' = 0$, ist per Definition vorgegeben, dass α_m den Wert $-V_F = \varpi - \pi$ hat (vgl. Abb. 2 und Tab. 1). Damit ist die Rektaszension der Vergleichssonne (mittlere Rektaszension) auf

$$\alpha_m = M - V_F \quad (8)$$

festgelegt. Darin bezeichnet

$$M = n_{tr} t' \quad (9)$$

die mittlere Anomalie und

$$n_{tr} = \frac{2\pi}{T_{tr}} \quad (10)$$

die mittlere Winkelgeschwindigkeit des Sonnenumlaufs.⁸ Man beachte, dass das (negative) konstante Glied $-V_F$ der mittleren Rektaszension – wie α_m – ein Winkel in der Äquatorebene ist, der aber nach Abb. 2 aus der Ekliptikebene stammt. Die Wahl von $-V_F$ als konstantes Glied gestaltet die Zeitgleichung g über die Umlaufperiode T_{tr} mittelwertfrei.

⁷ Die anderen Jahreszeitanfänge eigneten sich ebenso als Bezugszeit. Der sich aufdrängende Zeitpunkt der Perihelpassage ist ungeeignet, weil er stark vom Mond beeinflusst wird, der im Kepler-Modell fehlt. Der Mond kann die Spanne zwischen zwei Periheldurchgängen um Tage verschieben. Im Keplermodell ist sie konstant gleich T_{tr} . Die Alternative, einen Jahresanfang als Zeit-Nullpunkt zu wählen, sabotiert die einheitliche rechnerische Behandlung der Zeitgleichung über mehrere Jahre: Der Jahresanfang ist kein ausgezeichnete Punkt der Sonnenbahn.

⁸ Der Leser störe sich nicht an den geozentrischen Formulierungen. Der Autor hat die Kopernikanische Wende mitbekommen. Geozentrische Koordinaten sind weiterhin erlaubt.

6. Wahre Rektaszension

Zur Bestimmung der wahren Rektaszension α der Sonne für (1) betrachte man die grüne (ekliptische) und die blaue (äquatoriale) jeweils rechtwinklige Dreiecksfläche in Abb. 2. Im grünen Dreieck gilt $\ell_x = \ell \cos \Lambda$ und $\ell_y = \ell \sin \Lambda$, im blauen Dreieck $\ell_{y'} = \ell_y \cos \varepsilon$. Daraus folgt $\tan \alpha = \ell_{y'}/\ell_x = \ell \sin \Lambda \cos \varepsilon / (\ell \cos \Lambda) = \cos \varepsilon \tan \Lambda$. Die Rektaszension der Sonne erhält man durch Auflösung der letzten Gleichung zu

$$\alpha = \arctan_{\Lambda} (\cos \varepsilon \tan \Lambda). \quad (11)$$

Der Lageparameter-Index Λ an der arctan-Funktion besagt, dass hier *der* Zweig der [Arkustangens]-Relation aufzusuchen ist, dessen Wert dem – durch den Index bezeichneten – Lageparameter am nächsten kommt. Ein stetiger Λ -Verlauf wird dadurch sprunghaft auf α abgebildet. Λ ist gleichzeitig Argument und Lageparameter der Arkustangensfunktion.

7. Zeitgleichung

Die Zeitgleichung ergibt sich aus (1) mit (8) und (11) zu

$$g = n_{tr} t' - V_F - \arctan_{\Lambda} (\cos \varepsilon \tan \Lambda). \quad (12)$$

Die letzte Gleichung liefert g als Winkel in Bogenmaß. Wenn man, was meistens der Fall ist, die entsprechende Zeit in Minuten verlangt, ist die rechte Seite mit $24 \cdot 60 / (2\pi)$ zu multiplizieren.

Es verbleibt die Aufgabe, die P-Zeit t' und die ekliptische Länge Λ für das gegebene Datum D zu bestimmen.

8. Synchronisation mit dem Kalender

Man erhält die dem gegebenen Datum D zugehörige Zeit ab Frühlingsanfang nach (3).

Im Kepler-Modell beträgt die wahre Anomalie des Frühlingsanfangs (vgl. Abb 2)

$$V_F = \pi - \varpi. \quad (13)$$

Mit V_F kann die entsprechende exzentrische Anomalie aus

$$E_F = 2 \arctan_{\frac{V_F}{2}} \left(\sqrt{\frac{1-e}{1+e}} \cdot \tan \frac{V_F}{2} \right) \quad (14)$$

berechnet werden. Zum Index $\frac{V_F}{2}$ der arctan-Funktion gelten die Anmerkungen zu (11) sinngemäß.

Die P-Zeit bei Frühlingsanfang erhält man mit der Gleichung für die mittlere Anomalie

$$M = n_{tr} t' \quad (15)$$

und der Kepler-Gleichung

$$M = E - e \sin E, \quad (16)$$

indem man in (15) t'_F für t' und in (16) E_F für E einsetzt, die beiden rechten Seiten gleichsetzt und die neue Gleichung nach t'_F auflöst. Man erhält so

$$t'_F = \frac{1}{n_{tr}} (E_F - e \sin E_F). \quad (17)$$

Zeitdifferenzen hängen generell nicht vom Nullpunkt der Zeitskala ab, so dass aus $t' - t'_F = d(D) - d(D_F) = t$ (siehe (3))

$$t' = t + t'_F \quad (18)$$

für die Zeit ab Perihel folgt. Gl. (18) übersetzt die zum Datum D gehörige Skalenzeit t in die entsprechende P-Zeit des Kepler-Modells.

9. Exzentrische und wahre Anomalie

Mit t' lassen sich jetzt die mittlere Anomalie nach (15), die exzentrische Anomalie als Näherungslösung der Kepler-Gleichung (16) in Form der [\[Fourier-Teilsumme\]](#)

$$E = M + \left(e - \frac{e^3}{8}\right) \sin M + \left(\frac{e^2}{2} - \frac{e^4}{6}\right) \sin(2M) + \frac{3}{8}e^3 \sin(3M) + \frac{e^4}{3} \sin(4M) \quad (19)$$

und die wahre Anomalie

$$V = 2 \arctan \frac{E}{2} \left(\sqrt{\frac{1+e}{1-e}} \cdot \tan \frac{E}{2} \right) \quad (20)$$

zum Berechnungsdatum D angeben. Zum Index $\frac{E}{2}$ der arctan-Funktion gelten die Anmerkungen zu (11) sinngemäß. Die Näherungslösung nach (19) ist genauer als nötig.

10. Ekliptische Länge

Die ekliptische Länge bei D ergibt sich schließlich nach Abb. 2 zu

$$\Lambda = V - V_F. \quad (21)$$

Damit liegen alle Größen zur Ausrechnung der Zeitgleichung g nach (1) oder (12) vor.

11. Genauigkeit

Die Genauigkeit der Rechnung ist vor allem durch die Vereinfachungen des Kepler-Modells und weniger durch die Genauigkeit seiner Parameter (Nrn. 2 bis 9 in Tab. 1) begrenzt. Das Modell erfasst die Bahnstörungen durch Mond und Planeten nicht.

Im Zahlenbeispiel von Tab. 1 beträgt die Abweichung gegenüber dem Wert laut Computer-Almanach des US Marine-Obsevatoriums [\[MICA\]](#)⁹ 0,9 s. Wenn man einen anderen – auch weit entfernten – Frühlingsanfang verwendet, ändert sich das Ergebnis nicht, was sich durch die Konstanz der Umlaufperiode T_{tr} erklärt.

Im Zeitraum 1800 bis 2050 beträgt der maximale und mittlere Abweichungsbetrag (Abtastperiode 10 Tage) 2,5 bzw. 0,6 s. Man darf aufgrund der (hier nicht dargestellten) zeitlichen Verteilung der Abweichungen vermuten, dass die angegebenen Grenzen auch bis 2200 eingehalten werden, was aber mangels Referenzmodell nicht belegt ist.

12. Literatur und Web-Links

↑ [\[Arkustangens\]: Arkustangensfunktion mit Lageparameter](#)

↑ $[e_2, \dot{e}_2, \varpi_2]$: [E. M. Standish, JPL/Caltech: Keplerian Elements for Approximate Positions of the Major Planets, Table 1](#)

↑ $[\varepsilon_2, \dot{\varepsilon}_2]$: [A. Guthmann: Einführung in die Himmelsmechanik und Ephemeridenrechnung. 1994, S. 160.](#)

↑ [\[Fourier-Teilsumme\]: H. Soffel: Einführung in die Himmelsmechanik, S. 15, Gl. 31](#)

↑ [\[Kepler-Gleichung\]: Wiki Kepler-Gleichung](#)

↑ [\[Koordinatensysteme\]: Wiki Astronomische Koordinatensysteme](#)

↑ [\[MICA\]: U.S. Naval Observatory Multiyear Interactive Computer Almanac 1800-2050, Vers. 2.0](#)

↑ [\[Montenbruck\]: Grundlagen der Ephemeridenrechnung, Spektrum Akademischer Verlag Heidelberg, 7. Aufl. 2005, S. 139](#)

↑ [\[Schneider\]: Himmelsmechanik, Band II: Systemmodelle, BI-Wissenschaftsverlag, 1993, S. 507](#)

↑ [\[Scilab\]: Open source software for numerical computation](#)

↑ $[T_{tr}]$: [Observatoire de Paris: Eléments orbitaux kepleriens de l'orbite terrestre](#)

↑ $[\dot{\varpi}_2]$: [Wiki Apsidendrehung](#)

⁹ Der Almanach beziffert die Unsicherheit seiner Winkelangaben mit 0,01 bis 1 Bogensekunde, was bei der Zeitgleichung 0,7 bis 70 Millisekunden entspricht.




Abbildung 3: Elliptische Erdbahn in der Ekliptikebene um die Sonne im Brennpunkt
 Die Marken F, S, H und W stehen für die Jahreszeit-Anfänge auf der Nordhalbkugel, P und A für Perihel bzw. Aphel. Weitere Bezeichnungen siehe Tab. 1.

A. Übergang vom helio- zum geozentrischen Bezugssystem

In der geozentrischen Darstellung Abb. 2 taucht die wahre Anomalie V auf, eine Winkelgröße des *heliocentrischen* Kepler-Modells. Das irritiert auf den ersten Blick. Wie gelangt man vom heliozentrischen (ab der kopernikanischen Wende plausibleren) Modell nach Abb. 3 zum geozentrischen? Dahin führen drei Schritte:

- Sonne und Erde vertauschen, womit die Sonne die Erde umläuft
- Richtungspfeil(e) \Uparrow zum Frühlingspunkt um 180° drehen, wodurch die Sonne bei Frühlingsanfang von der Erde aus gesehen vor dem Frühlingspunkt steht
- Die heliocentrische Länge λ (der Erde) nach Kürzung ihres Bogenpfeils um 180° in die geozentrische Länge Λ (der Sonne) umbenennen, so dass die Länge, wie im heliozentrischen Kontext, ab Frühlingspunkttrichtung gemessen wird.

Das geozentrische Bild ist ebenso richtig wie das heliozentrische, solange man nur Entfernungen, Winkelgeschwindigkeiten und Winkel in Funktion der Zeit beschreibt. Für die Analyse der Zeitgleichung ist es geeigneter als das heliozentrische.

B. Programm-Code

Das folgende [Scilab]-Skript berechnet für ein gegebenes Datum D den Zeitgleichungswert g . Das Skript soll helfen, eventuelle Unklarheiten zu Details der Rechnung zu beseitigen. Alle Winkelgrößen sind in rad (Bogenmaß) ausgedrückt.

```
function [g,gGrd,gMin]=!g(D)
// Die Funktion !g(D) berechnet die Zeitgleichung für das Datum D.
// D kann in den Jahren 1800 - 2200 liegen. Die maximale Abweichung
// gegenüber einer genaueren Referenzmethode beträgt ca. 2.5 Sekunden.
```

```

// Eingabe:
// D: Datum, für das die Zeitgleichung g berechnet werden soll, Form [J,M,T,h,min,s]
// Ausgabe:
// g: Zeitgleichung/rad
// gGrd: Zeitgleichung/°
// gMin: Zeitgleichung/min
// Rechnung:
rad=%pi/180// Umrechnungsfaktor Grad->Bogenmaß
Jtr=365.24219// Umlaufperiode, tropisches Jahr (fester Parameter 1)
DF=[2000,3,20,7,35,0]// Frühlingsanfang 2000, UTC (fester Parameter 2)
t=datenum(D)-datenum(DF)// Umformung von D in Skalenzeit
t2=datenum(2000,1,1,12,0,0)-datenum(DF)// Standard-Epoche J2000.0 in Skalenzeit
T=(t-t2)/36525// Jahrhundert-Zeit für säkulare Änderungen von e, eps und varpi
e=0.016709 - 3.661e-5*T// numerische Exzentrizität, feste Parameter 3 & 4
eps=(23 + 26/60 + 21.4/3600 - 46.8/3600*T)*rad// Schiefe, feste Parameter 5 & 6
varpi=(102.93735 + 1.72*T)*rad// helioz. Länge des Perihels, feste Parameter 7 & 8
ntr=2*%pi/Jtr// mittlere Winkelgeschwindigkeit des Sonnenumlaufs
VF=%pi-varpi// wahre Anomalie bei DF
EF=2*!atan3(sqrt((1-e)/(1+e)).*tan(VF/2),ones(VF),VF/2)// exzentrische Anomalie bei DF
t_F=(EF-e.*sin(EF))/ntr// DF gemessen ab Perihel ("P-Zeit")
t_ =t+t_F// D als P-Zeit
M=ntr*t_// mittlere Anomalie bei D
alphan=M-VF// mittlere Rektaszension bei D
// exzentrische Anomalie bei D, Fourier-Teilsumme:
E=M+(e-e.^3/8).*sin(M)+(e.^2/2-e.^4/6).*sin(2*M)+3/8*e.^3.*sin(3*M)+e.^4/3.*sin(4*M)
V=2*!atan3(sqrt((1+e)/(1-e)).*tan(E/2),ones(E),E/2)// wahre Anomalie bei D
Lambda=V-VF// geoz. Länge bei D
alpha=!atan3(cos(eps).*tan(Lambda),ones(Lambda),Lambda)// Rektaszension der Sonne bei D
g=alphan-alpha// Zeitgleichung in rad
gGrd=g/rad// Zeitgleichung in Grad
gMin=gGrd*4// Zeitgleichung in Minuten
endfunction

function w=!atan3(y, x, wNah)// Arcustangens-Wert nächst Lageparameter wNah
// y: Ordinate, Typ Matrix
// x: Abszisse, Typ wie y
// wNah: Winkel, Skalar oder Typ wie y
// w: Winkel, Typ wie y
w=atan(y,x)
w=w+%pi*round( (wNah-w)/%pi )
endfunction

// Aufruf der Funktion:
D=[2000,2,1,12,30,45]// Datum, für das g zu berechnen ist, UTC
[g,gGrd,gMin]=!g(D)

```

C. Berechnungsvariante mit Zeitbezug aus mittlerer Länge

Im oben benutzten Modell definiert der Frühlingsanfang 2000 einen ausgezeichneten Punkt der Keplerbahn beim gegebenen Kalenderdatum D_F . Dieser Zeitbezug ist anschaulich, aber auch unüblich. Da der größte Teil des Rechenwegs auf die Bestimmung des Sonnenstands (heliozentrisch: der Erdposition) entfällt, liegt es näher, wie in der Ephemeridenrechnung üblich zu verfahren. Diese liegt dem Rechengang in Tab. 2 zugrunde. Es werden die in [Montenbruck] angegebenen mittleren Erdbahnelemente verwendet (Tabellenzeilen 2 bis 5). Der Zeitbezug wird durch das Bahnelement *mittlere Länge* L eingebracht (Zeile 4). Die mittlere Länge L wächst etwas schneller¹⁰ als die mittlere Anomalie M und wird vom Frühlingspunkt aus gemessen (Zeile 6). Die Keplergleichung wird mit

¹⁰Auf den kleinen Unterschied kommt es an.

Hilfe der Reihenentwicklung der Mittelpunktsgleichung C (Zeile 7) näherungsweise - aber genauer als nötig - gelöst. Die exzentrische Anomalie E erscheint deshalb im Rechengang nicht.

Das Rechenmodell der Tabelle 2 benötigt, wie das obige, 8 Konstanten (4 Werte zur Epoche J2000 und deren zeitliche Änderungen, Zeile 2 bis 5). Es liegt dem Scilab-Skript am Schluss von Anhang C zugrunde.

Berechnung der Zeitgleichung

Tabelle 2:

- 1 t = Zeit in Tagen, für die der ZG-Wert berechnet wird, gerechnet ab 1.1.2000 12:00 UT
- 2 e = $0.016709 - \frac{0.000042}{36525} t$
numerische Exzentrizität der Erdbahn (Bahnelement)
- 3 ϖ = $102.9400 + \frac{1.7192}{36525} \cdot t$
Länge des Perihels der Erdbahn in Grad (Bahnelement)
- 4 L = $100.4656 + \frac{36000.7690}{36525} \cdot t$
mittlere Länge der Erde in Grad (Bahnelement)
- 5 ε = $23 + 26/60 + 21/3600 - \frac{46.82/3600}{36525} \cdot t$
mittlere Schiefe des Äquators gegenüber der Ekliptik in Grad
- 6 M = $L - \varpi$
mittlere Anomalie der Erde in Grad
- 7 C = $\frac{180}{\pi} \left[\left(2e - \frac{e^3}{4} \right) \sin M + \frac{5}{4} e^2 \sin(2M) + \frac{13}{12} e^3 \sin(3M) \right]$
Mittelpunktsgleichung der Erde in Grad (Näherung genauer als nötig)
- 8 Λ = $L + C - 180$
geozentrische ekliptikale Länge der Sonne in Grad
- 9 α = $\arctan_{\Lambda}(\tan \Lambda \cos \varepsilon)$
Rektaszension der Sonne in Grad. Die Länge Λ im Index des Arkustangens ist der Schätzwert für den Nebenwert (Lageparameter).
- 10 α_M = $L - 180$
mittlere Rektaszension der Sonne in Grad
- 11 ZG = $4(\alpha_M - \alpha)$
Zeitgleichung in Zeitminuten

Scilab-Skript

```
clear, mode(0), lines(0), clc()
function [tD,ZG]= !ZG(DA, N, dt, DD)//Zeitgleichung im Keplerschen Bahnmodell
//mittlere Bahnelemente nach Montenbruck
//Zeitbezug per mittlerer Länge
//Lösung der Keplergleichung per Mittelpunktsgleichung
//Achtung: Gradmaß!
//Aufgerufene Skript-eigene Funktionen: Arkustangens mit Lageparameter !atan3
//Anwendbar von 1800 bis 2200
//Modellfehler: plusminus 2,5 s
```

```

//Eingabe:
//DA: Anfangstermin der ZG-Berechnung, z. B. DA=[1946 12 31 11 55 0]
//N: Anzahl Werte
//dt: Schritt in Tagen
//DD: wählbarer Zeitnullpunkt für Ergebnisdarstellung, z. B. DA oder
//[JJJJ 1 1 0 0 0] oder anders, beeinflusst ZG nicht

//Ausgabe
//tD: Darstellungszeit, Tage nach DD
//ZG: Zeitgleichung/min

grd=(%pi/180); //Umrechnungsfaktor ° -> rad
//Der Scilabbefehl datenum() wandelt ein Kalenderdatum in einen Zeitwert in
//Tagen (Scilab-Zeit) wie das Julianische Datum, aber mit anderem Nullpunkt.
tSc2000=datenum( [2000 1 1 12 0 0] ); //Scilab-Zeit/d der Epoche J2000.0
tScD=datenum(DD)//Scilab-Zeit/d des Nullpunkts der Darstellungszeit tD
tSc=datenum(DA)+dt*[0:N-1]'; //Termin(e)/d in Scilab-Zeit,
//Spaltenvektor oder Skalar
t=(tSc - tSc2000); // Zeitpunkt(e)/d der Termin(e) ab J2000.0
tD=tSc-tScD//Darstellungszeit/d, Tage nach DD
e=0.016709 - 0.000042/36525*t; //numerische Exzentrizität (Bahnelement der Erde)
varpi=102.9400 + 1.7192/36525*t; //Länge des Perihels/° (Bahnelement der Erde)
L=100.4656 + 36000.7690/36525*t; //mittlere Länge/° (Bahnelement der Erde)
eps=23 + 26/60 + 21/3600 -46.82/3600/36525*t; //Schiefe des Äquators/°,
M=L-varpi; //mittlere Anomalie/°
C=1/grd*( (2*e-1/4*e.^3).*sind(M) + 5/4*(e.^2).*sind(2*M) + 13/12*(e.^3).*sind(3*M) );
//Mittelpunktsgleichung/°
lambda=L+C-180; //geoz. ekliptikale Länge der SONNE/°
alpha=!atan3( tand(lambda).*cosd(eps),1,lambda*grd)/grd;
//Rektaszension der SONNE/°
alphaM=L-180; //mittlere Rektaszension der Sonne/° gleich mittlere Länge der Sonne
ZG=4*(alphaM - alpha); //Zeitgleichung/Minuten
endfunction

function w=!atan3(y, x, wNah)//Arcustangens-(Neben-)Wert nächst Lageparameter wNah
//y: Ordinate, Typ Matrix
//x: Abszisse, Typ wie y oder Skalar
//wNah: Winkel, Typ wie y oder Skalar
//w: Winkel, Typ wie y
x=x.*ones(y)//für skalares x
w=atan(y,x)
w=w+%pi*round( (wNah-w)/%pi )
endfunction

// Zeitgleichung für den 2. April 2015, 12:00 UT
[tD,ZG]= !ZG([2015 4 2 12 0 0], 1, 1, [2015 4 2 12 0 0])

xdel();
//Zeitgleichung für 5 Jahre, jeweils ab 1. Jan. 0:00 UT dargestellt
for Jahr=2012:2016
    [tD,ZG]= !ZG([Jahr 01 01 0 0 0], 366,1, [Jahr 1 1 0 0 0]);
    plot(tD,ZG,tD,zeros(tD))
end
xtitle('Zeitgleichung','Zeit/d jeweils ab 1. Januar 0:00 UT','ZG/min')

```